# $Id: PKGBUILD 87583 2013-04-03 13:18:34Z stativ $
# Maintainer: Lukas Jirkovsky <l.jirkovsky@gmail.com>
pkgname=klavaro
pkgver=1.9.7
pkgrel=1
pkgdesc="Free touch typing tutor program"
arch=('i686' 'x86_64')
url="http://klavaro.sourceforge.net/"
license=('GPL2')
depends=('curl' 'gtkdatabox' 'hicolor-icon-theme')
makedepends=('intltool')
optdepends=('espeakup: voice output')
install=$pkgname.install
source=(http://downloads.sourceforge.net/$pkgname/$pkgname-$pkgver.tar.bz2)
md5sums=('e28114568ac20f64b372f5af53abc5a3')

build() {
  cd "$srcdir"/$pkgname-$pkgver
#  CFLAGS="-lgmodule-2.0" ./configure --prefix=/usr
  ./configure --prefix=/usr
  make
}

package() {
  cd "$srcdir"/$pkgname-$pkgver
  make DESTDIR="$pkgdir" install

  sed -i 's|/usr/share/icons/hicolor/24x24/apps/klavaro.png|klavaro|' \
    "$pkgdir"/usr/share/applications/$pkgname.desktop
}
